

## Лабораторная работа №2

Кендысь Алексей, 3 курс, 7а группа

### Условие задачи

#### Вариант 8

Импресарио готовит выставку старинных автомобилей, среди которых могут быть Bugatti, Cadillac, Cobra, Corvette, Pierce Arrow, Studebaker. Опрос показал, что посмотреть именно Bugatti придут 58 специально приглашенных гостей, Cadillac — 37, Cobra — 42, Corvette — 40, Pierce Arrow — 55 и Studebaker — 33. Бюджет организации выставки составляет 15 млн у.е. Стоимость доставки автомобиля на выставку и обеспечение его сохранности составляют 6, 4, 3.8, 4.2, 5.5 и 3.2 млн у.е. соответственно. Задача импресарио в том, чтобы привлечь как можно больше специально приглашенных гостей, не превышая бюджет на организацию. Кроме того, на выставке должно быть не менее трех старинных автомобилей. Если Corvette будет выбран для выставки, то и Cobra должен там быть. Если же Bugatti отсутствует, то обязательно нужно включить в показ Cadillac. Постройте математическую модель и найдите оптимальное решение задачи. Определите, каким может быть минимальный и максимальный бюджет, чтобы выставка состоялась.

### Математическая модель

В задаче дано:

$n$  – число машин;

$g_i$  – количество приглашенных гостей для  $i$ -ой машины;

$c_i$  – стоимость доставки и обеспечения сохранности  $i$ -ой машины;

$cost\_max$  – бюджет организации;

$car\_min$  – минимальное количество машин на выставке;

В нашем случае имеем  $n = 6$  машин: Bugatti, Cadillac, Cobra, Corvette, Pierce Arrow, Studebaker.

Введём переменные модели:

$$x_i = \begin{cases} 1, & \text{если машина } i \text{ присутствует на выставке} \\ 0, & \text{в противном случае} \end{cases}$$

Также требуется учесть две импликации. Для этого используются первое и второе правило моделирования логических отношений:

#### Первое правило моделирования логических отношений

Пусть  $I$  — конечное множество индексов,  $x_i \in \{0, 1\}$ ,  $i \in I$  и  $0 \leq y \leq 1$ . Тогда импликация **если  $x_i = 0$  для всех  $i \in I$ , то  $y = 0$**  моделируется неравенством

$$y \leq \sum_{i \in I} x_i. \quad (1)$$

### Второе правило моделирования логических отношений

Пусть  $I$  — конечное множество индексов,  $I_0, I_1 \subseteq I$ ,  $I_0 \cap I_1 = \emptyset$ ,  $x_i \in \{0, 1\}$ ,  $i \in I$ , и  $y \in R$ ,  $0 \leq y \leq 1$ . Тогда импликация **если**  $x_i = 0$  для всех  $i \in I_0$  и  $x_i = 1$  для всех  $i \in I_1$ , **то**  $y = 0$  моделируется неравенством

$$y \leq \sum_{i \in I_0} x_i + \sum_{i \in I_1} (1 - x_i). \quad (3)$$

Импликация

**если**  $x_i = 0$  для всех  $i \in I_0$  и  $x_i = 1$  для всех  $i \in I_1$ , **то**  $y = 1$  моделируется неравенством

$$(1 - y) \leq \sum_{i \in I_0} x_i + \sum_{i \in I_1} (1 - x_i). \quad (4)$$

Тогда для импликации “Если Corvette выбран для выставки, то Cobra также должен быть выбран” получаем:

$$1 - x_{cobra} \leq 1 - x_{corvette}, \text{ либо:}$$

$$x_{cobra} \geq x_{corvette}$$

Для импликации “Если Bugatti отсутствует, то нужно включить в показ Cadillac”:

$$1 - x_{cadillac} \leq x_{bugatti}$$

Необходимо решить три задачи: максимизировать количество гостей, максимизировать бюджет, минимизировать бюджет.

Для первой задачи имеем модель:

$$\begin{aligned} & \max \left( \sum_{i=1}^n g_i \cdot x_i \right) \\ & \sum_{i=1}^n c_i \cdot x_i \leq cost\_max; \\ & \sum_{i=1}^n x_i \geq car\_min; \\ & x_{cobra} \geq x_{corvette}; \\ & 1 - x_{cadillac} \leq x_{bugatti}; \\ & x_i \in \{0, 1\}, \quad i = \overline{1, n}. \end{aligned}$$

Для второй задачи имеем модель:

$$\begin{aligned} & \max \left( \sum_{i=1}^n c_i \cdot x_i \right) \\ & \sum_{i=1}^n c_i \cdot x_i \leq cost\_max; \end{aligned}$$

$$\begin{aligned} \sum_{i=1}^n x_i &\geq car\_min; \\ x_{cobra} &\geq x_{corvette}; \\ 1 - x_{cadillac} &\leq x_{bugatti}; \\ x_i &\in \{0, 1\}, \quad i = \overline{1, n}. \end{aligned}$$

Для третьей задачи имеем модель:

$$\begin{aligned} \min &\left( \sum_{i=1}^n c_i \cdot x_i \right) \\ \sum_{i=1}^n c_i \cdot x_i &\leq cost\_max; \\ \sum_{i=1}^n x_i &\geq car\_min; \\ x_{cobra} &\geq x_{corvette}; \\ 1 - x_{cadillac} &\leq x_{bugatti}; \\ x_i &\in \{0, 1\}, \quad i = \overline{1, n}. \end{aligned}$$

### Листинг программы

Файл lab2.dat:

```
param n := 6;

param guests :=
1 58
2 37
3 42
4 40
5 55
6 33
;

param cost_max := 15;

param c :=
1 6
2 4
3 3.8
4 4.2
5 5.5
6 3.2
;

param car_min := 3;

param bugatti := 1;
param cadillac := 2;
param cobra := 3;

param corvette := 4;
```

Файл lab2.mod:

```
param n, integer, > 0;

set cars := 1..n;
```

```

param guests {cars}, > 0;
param cost_max, > 0;
param c {cars}, > 0;
param car_min, integer, >= 0;

param bugatti, integer, > 0, <= n;
param cadillac, integer, > 0, <= n;
param cobra, integer, > 0, <= n;
param corvette, integer, > 0, <= n;

var x {cars}, binary;

maximize guests_overall: sum{i in cars} guests[i] * x[i];
maximize max_total_cost: sum{i in cars} c[i] * x[i];
minimize min_total_cost: sum{i in cars} c[i] * x[i];

subject to cost: (sum{i in cars} c[i] * x[i]) <= cost_max;
subject to min_cars: (sum{i in cars} x[i]) >= car_min;
subject to implication1: x[cobra] >= x[corvette];

subject to implication2: (1 - x[cadillac]) <= x[bugatti];

```

Файл lab2.run:

```

reset;
model lab2.mod;
data lab2.dat;
option solver cplex;

objective guests_overall;
solve;
display guests_overall;
display x;
display min_total_cost;

objective max_total_cost;
solve;
display max_total_cost;
display x;
display guests_overall;

objective min_total_cost;
solve;
display min_total_cost;
display x;
display guests_overall;

```

## Результат

Задача 1:

```

CPLEX 22.1.1.0: optimal integer solution; objective 146
7 MIP simplex iterations
0 branch-and-bound nodes
guests_overall = 146

```

```

x [*] :=
1  1
2  0
3  0
4  0
5  1
6  1
;

```

```
min_total_cost = 14.7
```

Максимальное количество гостей – 146. При таком раскладе бюджет – 14.7.

Задача 2:

```
CPLEX 22.1.1.0: optimal integer solution; objective 14.7
3 MIP simplex iterations
0 branch-and-bound nodes
max_total_cost = 14.7
```

```
x [*] :=
1  1
2  0
3  0
4  0
5  1
6  1
;
```

```
guests_overall = 146
```

Максимальный бюджет – 14.7. При этом количество гостей – 146. Т.е. решение полностью совпадает с решением прошлой задачи, что вполне логично.

Задача 3:

```
CPLEX 22.1.1.0: optimal integer solution; objective 11
1 MIP simplex iterations
0 branch-and-bound nodes
min_total_cost = 11
```

```
x [*] :=
1  0
2  1
3  1
4  0
5  0
6  1
;
```

```
guests_overall = 112
```

Минимальный бюджет – 11. При этом количество гостей – 112.