

Лабораторная работа №1

Кендысь Алексей, 3 курс, 7а группа

Условие задачи

Вариант 67

Диетолог разрабатывает новую диету, состоящую из сливочного масла, натуральных бифштексов (мяса), хлеба и яблочного сока. Содержание калорий, белков, жиров, углеводов и холестерина (в 100 г продукта), а также максимальные и минимальные нормы их потребления (в день) приведены в таблице. Здесь же указана цена в рублях 100 г соответствующего продукта.

Содержание калорий, белков, жиров, углеводов и холестерина в продуктах

Элемент питания	Содержание в 100 г продукта				Норма потребления	
	масло	мясо	хлеб	сок	min	max
Калории	800	280	245	80	2 400	2 800
Белок	0.6 г	15 г	8 г	0 г	60 г	60г
Жир	20 г	5 г	0 г	0 г	0 г	30 г
Углеводы	0г	0г	5 г	Юг	Юг	40 г
Холестерин	0.15 г	0.08 г	0 г	0 г	0 г	0.5 г
Цена	60	350	25	50		

Определите дневной рацион, обеспечивающий получение необходимого количества питательных веществ при минимальных денежных затратах.

Математическая модель

В задаче дано:

n – число продуктов;

m – число элементов питания (питательных веществ);

c_j – цена j -го продукта (в руб., за 100 г продукта);

a_{ij} – содержание i -го питательного вещества в j -ом продукте (в 100 г продукта);

b_{\min_i} – минимальная норма потребления i -го питательного вещества;

b_{\max_i} – максимальная норма потребления i -го питательного вещества.

В нашем случае имеем $n = 4$ продукта: масло, мясо, хлеб и сок. А также имеем $m = 5$ питательных веществ: калории, белок, жир, углеводы и холестерин.

Введём переменные модели:

x_i – количество i -го продукта ($1 = 100$ г).

Необходимо минимизировать общее количество денежных затрат на продукты, при этом учитывая ограничения на нормы потребления основных питательных веществ. В итоге имеет модель следующего вида:

$$\min \left(\sum_{j=1}^n x_j c_j \right)$$
$$b_{\min_i} \leq \sum_{j=1}^n a_{ij} x_j \leq b_{\max_i}, \quad i = \overline{1, m};$$
$$x_j \geq 0, \quad j = \overline{1, n}.$$

Листинг программы

Файл lab1.dat:

```
param n := 4;  
param m := 5;
```

```
param c :=  
1 60  
2 350  
3 25  
4 50  
;
```

```
param a :=  
1 1 800  
1 2 280  
1 3 245  
1 4 80  
2 1 0.6  
2 2 15  
2 3 8  
2 4 0  
3 1 20  
3 2 5  
3 3 0  
3 4 0  
4 1 0  
4 2 0  
4 3 5  
4 4 10  
5 1 0.15  
5 2 0.08  
5 3 0  
5 4 0  
;
```

```
param b_min :=  
1 2400  
2 60  
3 0  
4 10  
5 0  
;
```

```
param b_max :=  
1 2800  
2 60  
3 30  
4 40  
5 0.5  
;
```

Файл lab1.mod:

```
param n, integer, > 0;  
param m, integer, > 0;
```

```
set foods := 1..n;  
set nutrients := 1..m;
```

```
param c {foods}, > 0;  
param a {nutrients, foods};  
param b_max {nutrients};  
param b_min {nutrients};
```

```

var x {foods}, >= 0;

minimize totalcost: sum{i in foods} c[i] * x[i];

subject to nutrition_min {i in nutrients}: b_min[i] <= sum{j in foods} (a[i, j]
* x[j]) <= b_max[i];

```

Файл lab1.run:

```

reset;
model lab1.mod;
data lab1.dat;
option solver cplex;
solve;

```

```
display totalcost;
```

```
display x;
```

Результат

```

CPLEX 22.1.1.0: optimal solution; objective 229.3299216
2 dual simplex iterations (0 in phase I)
totalcost = 229.33

```

```

x [*] :=
1  0.719655
2  0
3  7.44603
4  0
;

```

Получили минимальную стоимость в размере 229 руб., а также оптимальный ежедневный рацион – 71.9 г масла и 744.6 г хлеба.